

Personal  
Computer  
World  
CENTURY

NEW!

# 35 PROGRAMS FOR THE DRAGON 32



Dr Tim Langdell

# **35 PROGRAMS FOR THE DRAGON 32**

**Tim Langdell**



Copyright (c) Dr Tim Langdell 1982

*All rights reserved*

First published in Great Britain in 1982 by

Century Publishing Co. Ltd,

76 Old Compton Street, London W1V 5PA

in association with

Personal Computer World

ISBN 0 7126 0173 2 (paper)

Printed in Great Britain by  
Richard Clay (The Chaucer Press) Ltd,  
Bungay, Suffolk

# Contents

Picasso	..	..	..	..	..	..	..	7
Moire	..	..	..	..	..	..	..	8
Snowflake	..	..	..	..	..	..	..	9
Master Brain	..	..	..	..	..	..	..	10
Tweedledee	..	..	..	..	..	..	..	12
Runes	..	..	..	..	..	..	..	14
Spiral	..	..	..	..	..	..	..	16
Hi-Res Colour	..	..	..	..	..	..	..	17
Scateract	..	..	..	..	..	..	..	18
Weave	..	..	..	..	..	..	..	19
Concentration	..	..	..	..	..	..	..	20
Hangman	..	..	..	..	..	..	..	22
Reference	..	..	..	..	..	..	..	24
Mazenture	..	..	..	..	..	..	..	26
Microrite	..	..	..	..	..	..	..	30
Never a Cross Word	..	..	..	..	..	..	..	32
Horse Race	..	..	..	..	..	..	..	33
Calendar	..	..	..	..	..	..	..	34
Biorhythms	..	..	..	..	..	..	..	35
Web of Time	..	..	..	..	..	..	..	37
Kaleidoscope	..	..	..	..	..	..	..	38
3-Dee	..	..	..	..	..	..	..	39
Space Raider	..	..	..	..	..	..	..	40
Lunar Lander	..	..	..	..	..	..	..	42
Home Accounts	..	..	..	..	..	..	..	44
Character Creator	..	..	..	..	..	..	..	46
Musicollage	..	..	..	..	..	..	..	48
Organ Player	..	..	..	..	..	..	..	49
Hexapeek	..	..	..	..	..	..	..	50
Petrometer	..	..	..	..	..	..	..	52
Monitor	..	..	..	..	..	..	..	53
Metric Converter	..	..	..	..	..	..	..	55
Wumpii	..	..	..	..	..	..	..	56
Squash	..	..	..	..	..	..	..	58
Life	..	..	..	..	..	..	..	60



# Introduction

The Dragon 32 is probably the most sophisticated piece of equipment that you possess - certainly one of them - and yet without programs to run on it it is lifeless. This book was written to help you get started on your Dragon, to breath some life into it. Even if you have owned a Dragon for a while , this collection of games and useful programs should enable you to start getting the most from your micro. In fact, to give you some idea of what your Dragon is capable of, this entire book was produced using a Dragon linked to a high quality electronic typewriter.

I've included a number of action games like SPACE RAIDER and LUNAR LANDER as well as some thought provoking games such as MASTER BRAIN ,CONCENTRATION and WUMPII which all require a degree of skill to master. There are also some useful programs for general purposes such as a filing system (REFERENCE) and a HOME ACCOUNTS program. Last but not least there are a host of smaller programs designed to show off your Dragon's capabilities.

My intention in writing these programs was not to provide fully fledged programs that are 'perfect' of their kind - every program can be improved upon with a little thought. So I urge you to use these programs as a basis for your future programming, use them as stepping stones to more lengthy works. They contain many routines and 'tricks' which if you watch out will tell you many things about the Dragon which are not in the manual.

Being born in the Chinese year of the Dragon, this was a book I almost 'had' to do - and hope you enjoy the programs in it as much as I did writing them! I dedicate it with warm thoughts to my wife Cheri and daughter Melissa, and to anyone working as I have with Autistic children.



# Picasso

In Picasso I have made the Dragon draw random lines all over the screen. However, every now and again it will erase some lines by drawing a line PRESET instead of PSET.

```
10 REM *****
20 REM * PICASSO *
30 REM *****
40 PMODE 3,1
50 PCLS(2):SCREEN1,1
60 X=RND(255):Y=RND(191)
70 C=RND(2)
80 IF C=2 THEN 120
90 LINE -(X,Y),PSET
100 FOR DLAY=1 TO 200:NEXT
110 GOTO 60
120 LINE-(X,Y),PRESET
130 GOTO100
```



# Moire

When the Dragon 32 draws lines close to each other on the screen it does not draw them parallel to each other or at a predictable angle apart. Instead, drawing from a given point to alternate PIXEL positions around the edge of the screen causes a rather beautiful pattern. I've called it MOIRE after the Moire effect.

```
10 REM *****  
20 REM * MOIRE PATTERN *  
30 REM *****  
40 PMODE 4,1:PCLS:SCREEN1,1  
50 FOR X=0 TO 255 STEP 2  
60 LINE(128,98)-(0,X),PSET  
70 LINE (128,98)-(X,0),PSET  
80 LINE(128,98)-(255,X),PSET  
90 LINE (128,98)-(X,255),PSET  
100 NEXT  
110 GOTO 110
```

# Snowflake

The Dragon is capable of some very lovely graphic displays. Here is one called SNOWFLAKE.

Experimenting with new values for 'K' and 'T' can give rise to some interesting results. You might also try varying line 100 by replacing 8 with other values. This number will determine how many 'points' the shape will have.

```
10 REM *****
20 REM *SNOWFLAKE*
30 REM *****
40 PMODE 4,1
50 PCLS:SCREEN1,1
60 X=0:Y=6.284
70 J=360:K=20
80 T=(Y-X)/J
90 FOR P= X TO Y STEP T
100 N=K*COS(P*8)
110 V=N*SIN(P)
120 Z=N*COS(P)
130 PSET(128+V,96+Z,3)
140 NEXT P
150 K=K+10
160 GOTO 80
```

# Master Brain

MASTER BRAIN is a version of the ever popular game Master Mind. In it Dragon chooses at random a sequence of letters drawn from the letters A to G. You have to guess exactly which letters were chosen and in what order they are. Each sequence to be guessed is five letters long, and Dragon soon tells you if you try to enter more or less than this number of characters as your guess. If your guess contained a letter which was present at least once in his sequence then Dragon puts the letter L by your guess, one for each time it occurs. If you got both the letter and the position right, then Ps are put next to your guess - once for each letter in its correct position.

```
10 REM *****
20 REM * MASTER BRAIN *
30 REM *****
40 T$=" "
50 CLS
60 PRINT"I'M THINKING OF A SERIES OF
LETTERS MADE UP OF THE LETTERS 'A' TO
'G', WHAT ARE THEY?"
70 PRINT"PRESS 'S' TO START"
80 IF INKEY$="S" THEN GOTO 100
90 GOTO 80
100 CLS
110 FOR A=1 TO 5
120 X=RND(7)
130 A$=CHR$(X+64)
140 B$=B$+A$
150 NEXT
160 FOR G=1 TO 10
170 PRINT@400," "
180 PRINT@400
190 INPUT"YOUR GUESS";C$
200 IF LEN(C$)<>5 THEN GOTO 530
210 IFC$=B$ THEN 610
220 FOR Y=1 TO 5
```

```

230 P$=T$
240 T$=MID$(B$,Y,1)
250 IF T$=P$ THEN 280
260 F=INSTR(C$,T$)
270 IF F>0 THEN GOTO 570
280 NEXT Y
290 PRINT@(32*G),C$
300 PRINT@(32*G+10)," ";
310 IF J=0 THEN 390
320 FOR W=1 TO J
330 PRINT"L";
340 NEXT W
350 FOR Y=1 TO 5: T$= MID$(C$,Y,1)
360 F=INSTR(B$,T$)
370 IF F>0 THEN GOSUB 590
380 NEXT Y
390 PRINT@(32*G+20)," ";
400 IF K=0 THEN 440
410 FOR W=1 TO K
420 PRINT"P";
430 NEXT W
440 J=0:K=0
450 NEXT G
460 CLS
470 PRINT" YOU'VE HAD 10 GOES"
480 PRINT" AND FAILED...!"
490 PRINT" IT WAS ":"B$
500 INPUT "ANOTHER GAME";Q$
510 IF LEFT$(Q$,1)="Y" THEN RUN
520 STOP
530 PRINT@(32*14),"NO, 5 LETTERS!"
540 FOR DLAY=1 TO 300: NEXT
550 PRINT@(32*14)," "
560 GOTO170
570 J=J+1
580 GOTO 280
590 IF F=Y THEN K=K+1
600 RETURN
610 PRINT" YES IT WAS ";B$
620 GOTO500

```

# Tweedledee

In TWEEDLEDEE you have to remember a sequence of letters which Dragon puts up on the screen. He puts just two at first, then three, and so on up to ten. You simply type in what you thought you saw. As the game starts you will be asked what level of difficulty you want. '1' means easy, and '5' is almost impossible. You may know this game by one of its other names such as 'Simon'.

```
10 REM *****
20 REM *TWEEDLEDEE* 'LETTERS'
30 REM *****
40 CLS
50 P=0
60 INPUT"SPEED(1-5) ";X
70 FOR A=1 TO 10
80 FORDLAY=0TO 50:NEXT
90 CLS
100 PRINT"GO:";A
110 PRINT"PRESS 'S' WHEN YOU'RE READY"
120 IF INKEY$="S"THEN GOTO 140
130 GOTO120
140 B$=""
150 FOR D=1 TO A+1
160 C$=CHR$(64+RND(26))
170 B$=B$+C$
180 PRINT@238,C$
190 FOR DLAY=1TO800/X:NEXT
200 CLS
210 NEXTD
220 INPUT "WHAT LETTERS DID YOU SEE";L$
230 IF L$=B$ THEN GOTO330
240 PRINT"NO,SORRY"
250 FOR DLAY=1 TO 200:NEXT
260 NEXT A
270 CLS
280 PRINT"YOU'VE HAD 10 GOES."
290 PRINT"AND YOU HAVE SCORED"; P "POINTS!"
300 INPUT "ANOTHER GO";Q$
```

```
310 IF LEFT$(Q$,1)="Y" THEN RUN
320 STOP
330 CLS
340 PRINT"THAT'S CORRECT!"
350 FOR DLAY=1 TO 300:NEXT
360 P=P+1
370 GOTO 260
```

# Runes

Dragon as a poet? In RUNES The Dragon prints to the screen stanzas of a poem specially written for us by C. Davis Langdell. This done Dragon begins to play with the words - producing stanzas of his own. Many find this program fascinating and enjoy leaving it running to watch the many permutations.

```
10 REM *****
20 REM *   RUNES  **
30 REM *****
40 SCREEN0,1
50 DIMA$(5)
60 DIMB$(5)
70 DIMC$(5)
80 DIMD$(5)
90 GOSUB 260
100 CLS
110 FOR N=1 TO 5
120 PRINT:PRINT:PRINT: PRINT: PRINT
130 PRINTA$(N):PRINTB$(N): PRINT C$(N):
    PRINTD$(N)
140 FORA=1 TO 5000:NEXT
150 CLS
160 SOUND200,5
170 NEXT N
180 N1=1+RND(4):N2=1+RND(4)
190 N3=1+RND(4):N4=1+RND(4)
200 PRINT:PRINT:PRINT:PRINT
210 PRINTA$(N1):PRINTB$(N2): PRINTC$(N3):
    PRINTD$(N4)
220 FOR A=1 TO 5000:NEXT
230 CLS:SOUND 200,5
240 GOTO 180
250 STOP
260 REM ASSIGN LINES
270 RESTORE
280 FOR X=1 TO 5
290 READA$(X)
300 NEXT X
```

```
310 FOR X=1 TO 5
320 READ B$(X)
330 NEXT X
340 FOR X=1 TO 5
350 READ C$(X)
360 NEXT X
370 FOR X=1 TO 5
380 READ D$(X)
390 NEXT X
400 DATA "I love gambling by the Pen",
        "Hunger drives the need to win",
        "Southern fire,Northern Cross", "Leave
        your sadness here", "Come and bid the
        sunshine in"
410 DATA "The spring runs cold and clear",
        "June sees summer in", "Ripe grapes
        hanging in the park", "Rise before the
        lark", "Now the earth is young"
420 DATA "Blade by Blade, stem by stem",
        "Wit and magic can begin", "Persian cat
        under wolf moon lies", "Start the
        simple dance again","Fold the green
        leaf in your hand"
430 DATA "The wanderer asks,what cheer?",
        "Will it rain, silver fin?", "Who moves
        in the dark?", "Badger gnaws at the
        bark", "Touch earth sky, stars, sea, sun"
440 RETURN
```



# Spiral

Spiral is a program which does what you would expect it to do - draw a spiral - but having entered it play around with the various values to see the possible range of effects. I think you will be surprised at what your Dragon can do.

```
10 REM *****
20 REM * SPIRAL *
30 REM *****
40 PMODE 4,1
50 SCREEN1,1:PCLS
60 R=50
70 FOR F=0 TO 500
80 R=R-0.1
90 PSET(128+R* SIN(F/32*3.142),
      96+R*COS (F/32*3.142),1)
100 NEXT F
110 GOTO 110
```

# Hi-Res Colour

If you have experimented with your Dragon a little, then you will know that in the highest resolution mode you can only have two colours. However, here is a program which covers the screen with random colours in the form of small dots. Colour in high resolution! The latter half of the program is also interesting as it POKES the letters to the screen. Do not worry if this program is a little difficult to understand- it uses several tricks not revealed in the manual.

```
10 REM *****
20 REM *      HIGH      *
30 REM * RESOLUTION *
40 REM *    COLOUR    *
50 REM *****
60 X=RND(-TIMER)
70 POKE &HFFC3,0
80 POKE &HFFC5,0
90 FOR Y=1024 TO 7167
100 POKE Y,RND(128)+127
110 NEXT Y
120 FOR A=0 TO 352 STEP 32
130 POKE 3722+A,ASC("H")
140 POKE 3723+A,ASC("I")
150 POKE 3724+A,ASC("T")
160 POKE 3725+A,ASC("H")
170 POKE 3726+A,ASC("E")
180 POKE 3727+A,ASC("R")
190 POKE 3728+A,ASC("E")
200 NEXT A
210 GOTO 210
```

# Scateract

Here are two unusual graphics programs for the Dragon 32. In the first, SCATERACT, a harmless circle is painted using the PAINT command, and then stored in an array using GET. With the screen cleared the circle is placed back using PUT, but purposely using this feature wrongly for effect. I think you'll agree that it's interesting!

In the second, WEAVE, an idiosyncrasy of the Dragon's circle drawing facility is used to 'weave' many small circles together.

```
10 REM *****
20 REM * SCATERACT *
30 REM *****
40 DIMX(25,25)
50 PMODE 3,1
60 PCLS
70 SCREEN1,1
80 CIRCLE(128,90),25
90 PAINT (129,91),2,4
100 PAINT (129,92),3,4
110 GET (98,85)-(128,105),X,G
120 PCLS
130 FOR Y=1 TO 200 STEP 3
140 PUT (Y+30,85-Y/5)-(Y+55,105-Y/5),X,PSET
150 NEXT Y
160 FOR Y=50 TO 196 STEP 3
170 PUT (120+Y/5,Y)-(Y+33,100+Y/5), X,PSET
180 NEXT
190 GOTO 190
```

# Weave

```
10 REM *****
20 REM * WEAVE *
30 REM *****
40 PMODE 3,1
50 PCLS:SCREEN1,1
60 FOR X=1 TO 240 STEP 10
70 FOR Y=1 TO 170 STEP 10
80 CIRCLE(X,Y),10,,1,.3,.8
90 NEXT Y,X
100 GOTO 100
```

# Concentration

CONCENTRATION is an old favourite known by many names. In this game you are faced with four rows of four cards (yellow rectangles on the T.V. screen) and The Dragon will turn over a card at random to show what pattern is underneath. You then enter a number from 1 to 16 to indicate which card you wish to turn over. If the cards are the same then you get a point. You have 20 goes - good luck!

```
10 REM *****
20 REM *CONCENTRATION*
30 REM *****
40 CLS
50 LET G=1
60 DIM D(16)
70 DIM R(16)
80 S=0
90 FOR T=0 TO 16: D(T)=159: NEXT T
100 FOR A=100 TO 300 STEP 64
110 FOR X=1 TO 16
120 R(X)=RND(14)+224
130 NEXT X
140 GOSUB 320
150 J=RND(16)
160 CLS: D(J)=R(J)
170 N=1
180 GOSUB 320
190 INPUT "SQUARE"; P
200 D(P)=R(P)
210 GOSUB 320
220 IF R(P)=R(J) THEN S=S+1
230 FOR DLAY=0 TO 200: NEXT DLAY
240 LET G=G+1
250 IF G=20 THEN GOTO 270
260 GOTO 90
270 PRINT "YOU SCORED, "; S
280 INPUT "ANOTHER GO"; Q$
290 IF LEFT$(Q$,1)="Y" THEN GOTO 90
300 STOP
```

```
310 RUN
320 N=1
330 FOR A= 100 TO 300 STEP 64
340 FOR B=1 TO 16 STEP 4
350 PRINT@A+B,CHR$(D(N))+CHR$(D(N))
360 PRINT@A+B+31,N
370 N=N+1
380 NEXT B,A
390 RETURN
```

# Hangman

Here is a version of the popular game HANGMAN. The Dragon thinks of a word and you have ten tries to guess what he has thought of. If the letter you guess is right all occurrences of it in the word are put up on the screen. The number of dashes represents the number of letters in the word. As you guess, the man at the top of the screen will be drawn, and when he is complete you will have run out of chances.

```
10 REM *****
20 REM * HANGMAN *
30 REM *****
40 CLS
50 X=RND(7)
60 RESTORE
70 FOR Z=1 TO RND(20)
80 READ W$
90 NEXT Z
100 M(1)=208:M(2)=208:M(3)=214
110 M(4)=216:M(5)=212:M(6)=217
120 M(7)=218:M(8)=213:M(9)=217
130 M(10)=214
140 FOR Y= 1 TO LEN(W$)
150 PRINT@(32*6)+(Y*2),"- "; " "
160 NEXT Y
170 T=0
180 PRINT@(32*10),"GUESS!"
190 PRINT@(32*11),"
    "
200 F=0
210 INPUT L$
220 T=T+1
230 IF T=10 THEN 500
240 IF T>2 AND T<7 THEN 290
250 IF T>6 AND T<9 THEN 310
260 IF T>8 THEN 330
270 PRINT@(32*1)+10+T,CHR$(M(T))
280 GOTO 340
```

```

290 PRINT@(32*2)+7+T,CHR$(M(T))
300 GOTO 340
310 PRINT@(32*3)+4+T,CHR$(M(T))
320 GOTO 340
330 PRINT@(32*4)+T+2,CHR$(M(T))
340 REM CHECK ENTRY
350 IF L$=W$ THEN 550
360 FOR A= 1 TO LEN (W$)
370 IF L$=MID$(W$,A,1) THEN GOSUB 440
380 NEXT A
390 IF F=1 THEN 180
400 PRINT@(32*11),"SORRY,NO "
410 FOR DLAY=0 TO 400:NEXT
420 PRINT@(32*11)," "
430 GOTO 180
440 PRINT@(32*5)+(A*2),L$;
450 F=1
460 RETURN
470 DATA "FIELD","CARPET","MOUSE","HORSE",
"EXCHANGE","BUFFET","ALPHABET","RABBIT",
"COLOSUS","CRINGE","SHELF","PERPLEX",
"SALLOW","CONVEX","GALAXY","BOOK",
"ARTIFACT","LOVE","MINION","WORLD"
480 DATA 208, 208,
214, 216, 212, 217,
218, 213, 217, 214
490 REM lose
500 CLS:PRINT"SORRY, YOU LOSE"
510 PRINT"IT WAS ";W$
520 INPUT"ANOTHER GO?";Q$
530 IF LEFT$(Q$,1)="Y" THEN RUN
540 STOP
550 REM won
560 CLS:PRINT"YES IT WAS ";W$
570 INPUT"ANOTHER GO";Q$
580 GOTO 530

```



# Reference

REFERENCE is a very useful program which can be adapted for many purposes. You can label up to three columns (say, ROOM, BOOKCASE, SHELF, or REGION,SHOP,MANAGER, etc). In each column you can make up to 100 entries and the program makes it very simple to search for any given entry. Simply entering '^' for search allows you to say what is to be looked for and in which column. All three entries for that row are then displayed.

```
10 REM *****
20 REM * REFERENCE *
30 REM *****
40 CLS
50 F=0
60 CLEAR15000:DIMX$(100),Y$(100),Z$(100)
70 PRINT "THREE COLUMN HEADINGS"
80 INPUTA$,B$,C$
90 CLS
100 PRINT@0,A$;TAB(10);B$;TAB(20);C$
110 FOR X=1 TO 100
120 F=F+1
130 FOR H=0 TO 2
140 GOSUB220
150 IF H=0 THEN GOSUB 270
160 IF H=1 THEN GOSUB 290
170 IF H=2 THEN GOSUB 310
180 NEXT H
190 NEXT X
200 CLS:PRINT"THAT'S 100 ENTRIES"
210 STOP
220 PRINT@416,"":PRINT@384,"ENTRY?"
230 INPUT D$
240 IF D$="" THEN 330
250 PRINT@(32*X)+(H*10),D$;
260 RETURN
270 X$(F)=D$
280 RETURN
```

```

290 Y$(F)=D$
300 RETURN
310 Z$(F)=D$
320 RETURN
330 CLS
340 INPUT" LOOKING FOR";L$
350 PRINT"WHICH COLUMN(1,2,OR 3)?"
360 INPUT C
370 ON C GOTO 390,440,490
380 GOTO360
390 FOR G=1 TO 100
400 FOR R=0 TO 2
410 IF X$(G)=L$ THEN 540
420 NEXT:NEXT
430 GOTO610
440 FOR G= 1 TO 100
450 FOR R=0 TO 2
460 IF Y$(G)=L$ THEN 540
470 NEXT:NEXT
480 GOTO 610
490 FOR G=1 TO 100
500 FOR R=0 TO2
510 IF Z$(G)=L$ THEN 540
520 NEXT:NEXT
530 GOTO 610
540 CLS
550 PRINT@0,A$;TAB(10);B$;TAB(20);C$
560 PRINT@(32*3),X$(G);TAB(10);Y$(G);
TAB(20);Z$(G)
570 PRINT@416,"HIT ENTER TO CONTINUE"
580 INPUT U
590 CLS:PRINT@0,A$;TAB(10);B$;TAB(20)C$
600 GOTO 110
610 CLS
620 PRINT"SEARCH FAILED":PRINT:PRINT
630 PRINT" ANY KEY TO CONTINUE":PRINT"BREAK
TO END"
640 INPUT U
650 CLS:PRINT@0,A$;TAB(10);B$;TAB(20)C$
:GOTO 110

```

# Mazenture

MAZENTURE is a maze game and adventure in one. You start outside a maze and must find your way to the other side. But beware! There are many perils awaiting you, and at each encounter you must not only be carrying the correct item but also do the correct thing with it! Type 'N' for north, 'S' for south, etc. And other commands in simple English. 'M' lets you see the map and your position for a few moments, but only call it up to 5 times!

```
9 REM *****
10 REM * MAZENTURE *
11 REM *****
25 F=0
30 T=0
35 GOSUB1000
40 A=0:RESTORE
45 TRES=0
50 X=31
60 DIMZ$(36)
65 DIMI$(6)
70 DIMD$(1)
75 DIMS$(1)
80 FOR C=1 TO 36
90 READ Z$(C)
100 NEXT C
110 DATA "000110","001010", "000100",
"101010", "000110","001010"
120 DATA "100011","000111", "001010",
"000111", "001011","000001"
130 DATA "000101","001011", "000011",
"000011", "010111","001000"
140 DATA "000110"."001101", "011101",
"001011", "000101","001010"
150 DATA "000101","001010", "000110",
"011111", "001010","000011"
160 DATA "000100","001001", "000001",
"000101", "001101","101001"
```

```

165 CLS
170 INPUT " WHICH WAY";D$
180 PRINT@0, " "
190 IF TRES=1 AND D$="P" THEN I$
    (1+INT(X*6))="1"
200 IF D$="M" THEN GOTO 300
210 IF D$="N" THEN A=6
220 IF D$="S" THEN A=5
230 IF D$="E" THEN A=4
240 IF D$="W" THEN A=3
250 IF X=18 AND A=4 THEN 290
260 GOSUB 500
270 GOTO 170
290 CLS:PRINT"YOU'RE OUT!!"
295 STOP
300 REM MAP
305 CLS
310 PRINT@(32*5)+5, "***  ***  ***"
320 PRINT@(32*6)+5, " * * * * * "
330 PRINT@(32*7)+5, " * ***  ***  *"
340 PRINT@(32*8)+5, " * * * * * "
350 PRINT@(32*9)+5, "***  * * ***>>"
360 PRINT@(32*10)+5, "  * * * * "
370 PRINT@(32*11)+5, "*****  ***"
380 PRINT@(32*12)+5, " *      * ***"
390 PRINT@(32*13)+5, "***  *****  *"
400 PRINT@(32*14)+5, "  * * * * * "
410 PRINT@(32*15)+3, ">>***  * *****"
420 G=INT(X/6): H=X-1-(INT(X/6))*6
430 IF G<>(X/6) THEN 450
440 G=G-1:H=5
450 PRINT@(32*G*2)+132+(3*H), "O";
460 FOR DLAY=1 TO 600:NEXT:T=T+1
470 IF T=5 THEN GOTO 490
480 CLS:GOTO 170
490 CLS: PRINT@300,"YOU LOSE"
495 STOP
500 REM THIS WAY?
510 IF MID$(Z$(X),A,1)="0" THEN 540
520 IF MID$(Z$(X),A,1)="1" THEN X=X+(1 AND
    A=4)-(1 AND A=3)+(6 AND A=5)-(6 AND A=6)

```

```

525 IF MID$(Z$(X),1,1)="1" THEN 600
526 IF MID$(Z$(X),2,1)="1" THEN 700
530 RETURN
540 PRINT@0,"YOU CAN'T GO THAT WAY!":RETURN
550 STOP
600 REM TREASURE
605 CLS
610 PRINT" YOU CAN SEE A ";
611 IF INT(X/6)=1 THEN PRINTV$
612 IF INT(X/6)=0 THEN PRINTL$
613 IF INT(X/6)=6 THEN PRINTES$
620 INPUT "WHAT DO YOU WISH TO DO";S$
630 IF S$="P" THEN GOTO 640
635 PRINT@0,"OK DON'T HAVE IT ":RETURN
640 PRINT@0,"YOU HAVE IT NOW"
645 TRES=1
646 CLS
650 RETURN
699 STOP
700 REM TROUBLE
710 PRINT"THERE'S A ";
711 IF INT(X/6)=2 THEN PRINTQ$
712 IF INT(X/6)=3 THEN PRINTU$
713 IF INT(X/6)=4 THEN PRINTM$
720 INPUT "WHAT DO YOU DO";S$
730 IF S$=Y$(1) AND INT(X/6)=4 THEN 800
733 IF S$=Y$(2) AND INT(X/5)=4 THEN 800
740 PRINT@0,"SORRY ABOUT THAT"
741 PRINT"HE CAST A SPELL ON YOU"
742 STOP
800 IF TRES>0 THEN 820
810 PRINT"SORRY, THAT WILL NOT DO...":STOP
820 LET TRES=0: PRINT@0,"HE LIKED
    THAT PLEASE GO ON":PRINT@32," "
830 GOTO 170
1000 V$=" DIAMOND BRACELET"
1010 L$=" VOGON POEM"
1020 E$=" FRUIT BUN"
1030 Q$=" A DRAGON"
1040 M$=" MOUSE"
1050 U$=" SNAKE"

```

```
1060 DIM Y$(3)
1070 Y$(1) = "R" : Y$(2) = "O" : Y$(3) = "T"
1080 RETURN
```

# Microrite

MICRORITE is a mini wordprocessor. In its way it is quite powerful, allowing you to enter a page of text and edit that text at the press of a key. To edit you simply enter 'E' and then 'I' to insert a word/phrase or 'D' to delete a space. Entering 'I' and then ENTER puts a single space in the text.

```
10 REM *****
20 REM * MICRORITE *
30 REM *****
40 CLEAR10000
50 W=1
60 CLS
70 T$=""
80 FLAG=0:PRINT@0,T$
90 PRINT@(32*14)," ";
100 PRINT"YOUR SENTENCE?(BREAK TO STOP) "
110 INPUT S$
120 IF S$="E" THEN GOSUB 200
130 IF S$="P" THEN GOSUB 390
140 IF S$="H" THEN GOSUB 470
150 CLS
160 IF FLAG =1 THEN 80
170 T$=T$+S$
180 PRINT@0,T$
190 GOTO 90
200 REM EDIT
210 FLAG=1
220 PRINT@0,LEFT$(T$,W);CHR$(159);
    RIGHT$(T$,LEN(T$)-(W AND LEN(T$)>W))
230 A$=INKEY$
240 IF A$="." THEN W=W+(1 AND W<LEN(T$))
250 IF A$="," THEN W=W-(1 AND W>0)
260 IF A$="I" THEN GOSUB 300
270 IF A$="D" THEN GOSUB 420
280 IF A$="R" THEN RETURN
290 GOTO 220
300 REM INSERT
310 CLS
```

```

320 PRINT"WHAT DO YOU WISH TO INSERT"
330 INPUT I$
340 IF I$="" THEN I$=" "
350 B$=LEFT$(T$,W):C$=RIGHT$(T$,LEN(T$)-W)
360 T$="":T$=B$+I$+C$
370 CLS
380 RETURN
390 PRINT#-2," ";T$
400 FLAG=1
410 RETURN
420 REM DELETE
430 IF W=1 THEN RETURN
440 B$=LEFT$(T$,W-1):C$=RIGHT$(T$,LEN(T$)-
(W AND LEN(T$)>W))
450 T$="":T$=B$+C$
460 RETURN
470 REM HELP - MENU
480 FLAG=1
490 CLS
500 PRINT@(32*5)+10,"MENU"
510 PRINT@(32*7)," 'P' = SEND TO PRINTER"
520 PRINT:PRINT"'E' = EDIT TEXT"
530 PRINT:PRINT"'I' = INSERTION"
540 PRINT:PRINT"'D' = DELETION"
550 PRINT:PRINT"'R' = RETURN"
560 P$=INKEY$
570 IF P$="R" THEN RETURN
580 GOTO 560

```



# Never a Cross Word

In many magazines you will find word games of the kind which ask you to find out how many three letter words can be produced from one long one. NEVER A CROSS WORD does this for you, and in seconds. It randomly chooses three of the letters in the long word and prints them to the screen. When you've seen enough press BREAK to stop the program. You can see that it is equally easy to search for four, five or more letter words too.

```
10 REM *****
20 REM *   NEVER A   *
30 REM * CROSS WORD *
40 REM *****
50 CLS
60 PRINT"THIS PROGRAM IS DESIGNED"
70 PRINT"TO PRINT OUT ALL POSSIBLE"
80 PRINT"COMBINATIONS OF 3 LETTERS"
90 PRINT"FROM A LONG WORD."
100 INPUT"LONG WORD";W$
110 CLS
120 X=INT((LEN(W$)*RND(0))+1)
130 Y=INT((LEN(W$)*RND(0))+1)
140 IF Y=X THEN 130
150 Z=INT((LEN(W$)*RND(0))+1)
160 IF Z=Y OR Z=X THEN 150
170 PRINTMID$(W$,X,1);MID$(W$,Y,1);MID$(
    W$,Z,1);" ";
180 FOR DLAY=1 TO 100:NEXT
190 GOTO 120
```

# Horse Race

Place your bets! HORSE RACE shows a depiction of six horses racing each other, and you are invited to bet on one.

```
10 REM *****
20 REM * HORSE RACE *
30 REM *****
40 PRINT"WELCOME TO THE HORSE RACES"
50 PRINT"CARE TO BET ON A HORSE,1 TO 6??"
60 INPUT"YOU BET ON";B
70 CLS
80 FOR A=1 TO 15
90 PRINT@(32*A)+20,CHR$(157)
100 NEXT
110 DIMH(6)
120 FOR P= 1 TO 6
130 PRINT@((32*P)+H(P)), " ";
140 H(P)=H(P)+RND(2)
150 PRINT@((32*P)+H(P)),P;
160 SOUND 200,1
170 IF H(P)>15 THEN GOTO 200
180 NEXT P
190 GOTO 120
200 FORDLAY=1 TO 200: NEXT: CLS:PRINT"THE
    WINNER WAS HORSE ";P
210 IF P=B THEN GOTO 260
220 PRINT"PITY YOU DID NOT BET ON IT!"
230 INPUT"ANOTHER GO";Q$
240 IF LEFT$(Q$,1)="Y" THEN RUN
250 STOP
260 PRINT"AND YOU BET ON IT!"
270 GOTO 230
```

# Calendar

CALENDAR will calculate what day of the week a date fell upon for any day since the beginning of the Gregorian calendar. A potentially very useful program for forward planning, and of great use as a routine in longer programs too (see BIORHYTHMS).

```
5 REM *****
10 REM * CALENDAR *
20 REM *****
30 CLS
40 DIM K(12)
50 FOR J=1 TO 12
60 READ K(J)
70 NEXT J
80 PRINT "WHICH DATE (PUT DAY[2
  CHR$],MTH[2CHR$],YR[4CHR$])"
90 INPUT"DAY";D
100 INPUT"MONTH";M
110 INPUT"YEAR";Y
120 X=Y*365+INT(Y/4)+INT(Y/400)
  - INT(Y/100) - (M<3 AND Y/4=INT(Y/4)
  AND (Y/100=INT(Y/100) OR Y/400=INT
  (Y/400))) 130 X=X+K(M)+30 *(M-2)+D
140 Z=X-7*INT(X/7)+1
150 M$="SASUMOTUWETHFR"
160 PRINT MID$(M$,2*Z-1,2);" ";D;"/";M;"/";Y
170 INPUT "ANOTHER";Q$
180 IF LEFT$(Q$,1)="Y" THEN RUN
190 STOP
200 DATA 30, 31, 29, 30, 30, 31,
  31, 32, 33, 33, 34, 34
```

# Biorhythms

BIORHYTHMS will calculate where you are on each of a 33-day cycle of intellect, a 28-day cycle of emotion, and a 23-day cycle of vitality. It assumes that all these three cycles began at the day of your birth - although there is very little evidence for these cycles they can be fascinatingly accurate!

```
10 REM *****
20 REM * BIORHYTHMS *
30 REM *****
40 DIMK(12)
50 FOR J=1 TO 12
60 READK(J)
70 NEXT J
80 PRINT"ENTER YOUR DATE OF BIRTH"
90 PRINT"PUT 2 DIGITS FOR THE DAY"
100 PRINT"PUT 2 DIGITS FOR THE MONTH"
110 PRINT"PUT 4 DIGITS FOR THE YEAR"
120 PRINT"E.G. 03,11,1997"
130 GOSUB 240
140 PRINT" TODAY'S DATE? (SAME FORMAT) "
150 LET Q=X
160 GOSUB 240
170 X=X-Q
180 PRINT"INTELLECT";X-33*INT(X/33)
190 PRINT"EMOTION";X-28*INT(X/28)
200 PRINT"VITALITY";X-23*INT(X/23)
210 INPUT"ANOTHER";Q$
220 IF LEFT$(Q$,1)="Y" THEN RUN
230 STOP
240 INPUT"DAY";D
250 INPUT"MONTH";M
260 INPUT"YEAR";Y
270 X=Y*365+INT(Y/4)+INT(Y/400)-INT(Y/100)
      - (M<3 AND Y/4=INT(Y/4) AND (Y/100=INT
      (Y/100) OR Y/400=INT(Y/400)))
280 X=X+K(M)+30*(M-2)+D
290 Z=X-7*INT(X/7)+1
```

```
300 M$="SASUMOTUWETHFR"
310 PRINT MID$(M$,2*2-1,2);" ";D;"/";M;
    "/";Y
320 RETURN
330 DATA 30, 31, 29, 30, 30, 31, 31, 32,
    33, 33, 34, 34
```

# Web of Time

WEB OF TIME draws a network of latices on the screen in ever decreasing sizes. Many routines in this program could be the basis of more adventurous programs.

```
10 REM *****
20 REM * WEB OF TIME *
30 REM *****
40 PMODE 4,1
50 PCLS
60 SCREEN1,1
70 DIMX(36)
80 DIMY(36)
90 A=120
100 B=80
110 FOR J=1 TO 5
120 FOR K=1 TO 36
130 T=K/18*3.142
140 X(K)=128+(A*SIN(T))
150 Y(K)=88+(B*COS(T))
160 PSET(X(K),Y(K),1)
170 NEXT K
180 FOR K=1 TO 36
190 G=K+12
200 IF G>36 THEN G=G-36
210 P=X(G)-X(K)
220 Q=Y(G)-Y(K)
230 LINE(X(K),Y(K))-((X(K)+P),(Y(K)+Q))
240 NEXT K
250 A=A/2:B=B/2
260 NEXT J
270 GOTO 270
```

# Kaleidoscope

KALEIDOSCOPE is a simple graphics program that can be quite fascinating to watch. As its name suggests, it plots dots in a kaleidoscopic fashion on the screen. Why not add a subroutine to occasionally PRESET some dots to make the display more dynamic?

```
10 REM *****
20 REM * KALEIDOSCOPE *
30 REM *****
40 PCLS (1)
50 PMODE 3,1
60 SCREEN1,1
70 A=RND(127)
80 B=RND(96)
90 PSET(128+A,96+B,RND(4))
100 PSET(128-A,96-B,RND(4))
110 PSET(128+A,96-B,RND(4))
120 PSET(128-A,B+96,RND(4))
130 GOTO70
```

## 3-Dee

Your Dragon 32 is capable of quite sophisticated graphics , and 3-DEE is a very good example of high resolution plotting at its most interesting. The program plots the equation in line 70, so you might try making changes to this line and experiment with the effects produced.

```
10 REM *****
20 REM * 3-DEE *
30 REM *****
40 PMODE 4,1:PCLS:SCREEN1,1
50 FOR X=-100 TO 100
60 R=10:J=0:K=1
70 V=R*INT(SQR(10000-X*X)/R)
80 FOR Y=V TO -V STEP -R
90 Z=INT(80+30*SIN((SQR(X*X+Y*Y))/12)-0.7*Y)
100 IF Z<J THEN 140
110 J=Z
120 PSET(X+110,Z-15,1)
130 K=0
140 NEXT Y: NEXT X
150 GOTO 150
```



# Space Raider

The alien fleet is attacking and you alone are earth's defence. Your base is a pulsing laser-former at the bottom of the screen which is continuously firing rapid volleys of lethal light. Move your base with the '>' key. Notice how this program creates the alien graphic in lines 100 to 240 and how it PUTs it to the screen in successive positions. Note too that it PUTs a blank array over the last position to black it out (lines 310 and 330). The explosion is created by PUTting the alien to the screen in a smaller box than it needs. A very useful effect!

```
10 REM *****
20 REM * SPACE RAIDER *
30 REM *****
40 S=0
50 FORA=1 TO 8
60 G(A)=RND(255)
70 NEXT A
80 DIML(9,9)
90 DIMA$(9)
100 DIMC(9,9)
110 A$(1)="01011010"
120 A$(2)="00100100"
130 A$(3)="01111110"
140 A$(4)="01000010"
150 A$(5)="00100100"
160 A$(6)="10011001"
170 A$(7)="10000001"
180 A$(8)="10000001"
190 PMODE4,1:PCLS
200 FORX=1 TO 8 : FOR Y=1 TO 8
210 K=0: IF MID$(A$(X),Y,1)="1" THEN K=1
220 PSET(9+Y,9+X,K)
230 NEXT: NEXT
240 GET(10,10)-(18,18),C
250 PCLS
260 X=100
```

```
270 SCREEN1,1
280 D=8
290 FORJ=1 TO 255STEP 2
300 PUT (J,110)-(J+D,118),C
310 PUT (J,110)-(J+8,118),L
320 PUT (X,180)-(X+8,196),G
330 PUT (X,180)-(X+8,192),L
340 D=8
350 IF INKEY$="." THEN X=X+10
360 IF X>255 THEN X=0
370 IF X<0 THEN X=0
380 GOSUB 410
390 NEXT J
400 GOTO 290
410 REM
420 PSET (X,196-(P*10),1)
430 PSET (X,196-(P*10),0)
440 P=P+1:IF P=10 THEN P=0
450 IF P=9 AND X=J OR X=J-1 OR
    X=J+1 THEN S=S+1
460 R=R+1:IF R=400 THEN GOTO 490
470 IF P=9 AND X=J OR X=J-1 OR X=J+1
    THEN LET D=0
480 RETURN
490 SCREEN0,1
500 PRINT"YOU SCORED ";S;" POINTS"
510 STOP
```

# Lunar Lander

In LUNAR LANDER you are in total control of a space craft coming in to land. You adjust its rate of decent with the up arrow and down arrow cursor keys. If you judge its speed well and don't run out of fuel by thrusting too much, then you will land safely.

```
10 REM *****
20 REM * LUNAR LANDER *
30 REM *****
40 DIML(9,9)
50 DIMA$(9)
60 DIMC(9,9)
70 A$(1)="01011010"
80 A$(2)="00100100"
90 A$(3)="01111110"
100 A$(4)="01000010"
110 A$(5)="00100100"
120 A$(6)="10011001"
130 A$(7)="10000001"
140 A$(8)="10000001"
150 PMODE4,1:PCLS
160 FORX=1 TO 8 : FOR Y=1 TO 8
170 K=0: IF MID$(A$(X),Y,1)="1" THEN K=1
180 PSET(9+Y,9+X,K)
190 NEXT: NEXT
200 GET(10,10)-(18,18),C
210 PCLS
220 SCREEN1,1
230 A=1500
240 TH=0
250 F=20
260 V=0
270 REM DECENT
280 J=1509-A: K=J+8
290 PUT(110,J)-(118,K),C
300 PUT(110,J)-(118,K),L
310 LINE(0,170)-(255,170),PSET
320 A$=INKEY$
```

```
330 IF A$="[" THEN TH=TH+1
340 IF A$="]" THEN TH=TH-1
350 V=V+TH/10
360 IF V<0 THEN 380
370 A=A-V
380 F=F-TH/5
390 IF F<0 OR A<1350 THEN GOTO 410
400 GOTO 270
410 CLS:IF V<5 AND F>0 THEN GOTO 450
420 PRINT"YOU CRASHED WITH FUEL ";F
430 PRINT"AND AT A VELOCITY OF ";V
440 STOP
450 PRINT"YOU LANDED SAFELY!!"
460 PRINT"WITH FUEL TO SPARE AT ";F
470 PRINT" AND A GOOD VELOCITY OF 480
    STOP ";V
```

# Home Accounts

HOME ACCOUNTS will undoubtedly come in useful in putting your Dragon 32 to work in the home. With this program you can enter up to 100 debits or credits and see a running total/balance displayed. You could balance your cheque book with the program, or even assess the finances of a small business. By simply pressing 'P' in response to the prompt your entries will be sent to a printer if you have one attached. It would be equally easy to add one more routine to send them to your cassette recorder, but upon reloading be sure to use 'GOTO 10' and not RUN or your numbers will be lost!

```
10 REM *****
20 REM * HOME ACCOUNTS *
30 REM *****
40 DIMC(100):DIMD(100):DIMB(100)
50 CLS
60 X=0
70 PRINTTAB(5);"HOME ACCOUNTS"
80 PRINT:PRINT:PRINT"THIS
   PROGRAM WILL ALLOW YOU TO "
90 PRINT"ENTER A NUMBER OF OUTGOINGS
100 PRINT"A NUMBER OF INCOMES"
110 PRINT"PRESS ENTER TO CONTINUE"
120 INPUT Q$
130 CLS
140 GOTO 330
150 REM MAIN LOOP
160 PRINT:PRINT
170 INPUT"INCOME (I) OR OUTGO(O) OR PRINTER
   (P) ";R$
180 X=X+1
190 IF R$="I" THEN 230
200 IF R$="P" THEN 380
210 IF R$="O" THEN 280
220 GOTO 190
230 REM CREDIT
```

```

240 INPUT "AMOUNT";C
250 C(X)=C
260 B(X)=B(X-1)+C
270 GOTO320
280 REM DEBIT
290 INPUT" AMOUNT"; D
300 D(X)=D
310 B(X)=B(X-1)-D
320 CLS
330 PRINT"INCOME";TAB(8);"OUTGO";TAB(18);
    "BALANCE"
340 FOR A=1 TO X
350 PRINTC(A);TAB(8);D(A);TAB(18);B(A)
360 NEXT A
370 GOTO 150
380 REM PRINTER
390 PRINT#-2," "
400 PRINT#-2,"INCOME";TAB(8);"OUTGO";TAB
    (18);"BALANCE"
410 FOR A=1 TO X
420 PRINT#-2,C(A);TAB(8);D(A);TAB(18);B(A)
430 NEXT A
440 GOTO 320

```

# Character Creator

It is possible to create your own characters on the Dragon by using PSET in mode 4, and by using GET to store the character in an array. This program makes the creation of characters in this manner very simple. You just have to imagine each character as being built up of eight rows of eight dots. Simply taking a '1' to be a dot and a '0' as a space, enter each row in turn putting eight '0's and '1's after the prompt. When you've entered eight Dragon will print your character on the high-resolution screen using PSET, and then use GET to store it in array 'C'. You can then use PUT in your own program to PUT it anywhere on the screen. See the LUNAR LANDER and SPACE RAIDER programs to see this in action.

```
10 REM *****
20 REM * CHARACTER *
30 REM * CREATOR *
40 REM *****
50 CLS
60 PRINT TAB(4); "A CHARACTER CREATOR"
70 PRINT:PRINT"ENTER EACH LINE OF THE CHR$"
80 PRINT"AS A SERIES OF 8
   '0'S OR '1'S"
90 DIMA$(8)
100 PRINT" 12345678"
110 FOR N=1 TO 8
120 INPUT A$(N)
130 NEXT N
140 PMODE4,1
150 PCLS:SCREEN 1,1
160 DIMC(9,9)
170 FOR X=1 TO 8: FOR Y=1 TO 8
180 K=0: IF MID$(A$(X),Y,1)="1" THEN K=1
190 PSET(9+Y,9+X,K)
200 NEXT:NEXT
210 GET (10,10)-(18,18),C
```

```
220 PCLS
230 PUT(110,110)-(118,118),C
240 FOR DLAY=1 TO 1500:NEXT
250 CLS
260 PRINT"THE CHR$ IS NOW STORED IN ARRAY "
270 STOP
```



# Musicollage

MUSICOLLAGAGE is purely for fun. It prints various graphic characters to the screen, at random positions. At the same time it SOUNDS a random series of notes at random lengths. Can be quite fascinating!

```
10 REM *****
20 REM * MUSICOLLAGAGE *
30 REM *****
40 CLS
50 A=RND(30)
60 B=RND(15)
70 C=RND(255)
80 D=RND(5)
90 E=RND(50)+175
100 PRINT@(32*B)+A,CHR$(E);
110 SOUND C,D
120 GOTO 50
```

# Organ Player

ORGAN PLAYER turns your Dragon into a simple electronic organ. The keys Q to P are the notes F to A, and the numerical keys above have the sharps and flats just as on a piano. It would be quite easy to add differing octaves by detecting INKEY\$ and having O3 and O4 as variables. Or you could change the tempo by detecting another key press and using the 'T' instruction in the PLAY string.

```
10 REM *****
20 REM * ORGAN PLAYER *
30 REM *****
40 A$=INKEY$
50 IF A$="1" THEN PLAY"O3;5"
60 IF A$="Q" THEN PLAY"O3;6"
70 IF A$="2" THEN PLAY"O3;7"
80 IF A$="W" THEN PLAY"O3;8"
90 IF A$="3" THEN PLAY"O3;9"
100 IF A$="E" THEN PLAY"O3;10"
110 IF A$="4" THEN PLAY"O3;11"
120 IF A$="R" THEN PLAY"O3;12"
130 IF A$="T" THEN PLAY "O4;1"
140 IF A$="6" THEN PLAY"O4;2"
150 IF A$="Y" THEN PLAY"O4;3"
160 IF A$="7" THEN PLAY"O4;4"
170 IF A$="U" THEN PLAY"O4;5"
180 IF A$="I" THEN PLAY"O4;6"
190 IF A$="9" THEN PLAY"O4;7"
200 IF A$="O" THEN PLAY"O4;8"
210 IF A$="0" THEN PLAY"O4;9"
220 IF A$="P" THEN PLAY"O4;10"
230 GOTO 40
```

# Hexapeek

With HEXAPEEK and its sister program MONITOR you can both inspect the bytes of code in your Dragon's memory and enter your own machine code if you wish. HEXAPEEK asks you which address you want to start at and then displays line by line each memory location together with the decimal byte, the byte in HEX and then the CHR\$ of the byte. In this way you can either inspect the Dragon's internal Read Only Memory (ROM) or inspect an area where you have put code yourself. The display is rather fast so to help you see it I've added a line to detect the pressing of the SPACE BAR. When this is pressed the display pauses for a moment. Keep pressing it to get a slower rate of display. Press BREAK when you've seen enough or want to inspect a series of locations in more depth.

```
10 REM *****
20 REM * HEXAPEEK *
30 REM *****
40 CLS
50 PRINT"*****"
60 PRINT"          HEXAPEEK"
70 PRINT"*****"
80 PRINT:PRINT"THIS PROGRAM ALLOWS YOU TO
   LOOK AT THE MEMORY LOCATIONS IN YOUR
   DRAGON 32"
90 PRINT:PRINT"PRESS ENTER TO CONTINUE"
100 INPUTD$
110 L=4.55*G
120 PRINT:PRINT"IT WILL PRINT IN A LINE
   FIRST"
130 PRINT"THE ADDRESS, THEN THE DEC-"
140 PRINT"IMAL BYTE THERE, AND THEN THE"
150 PRINT"BYTE IN HEX, AND FINALLY "
160 PRINT"THE CHR$ OF THE BYTE"
170 PRINT"PRESS THE SPACE BAR TO SLOW"
180 PRINT"DOWN THE DISPLAY AT ANY TIME"
```

```

190 PRINT:PRINT"ENTER THE FIRST MEMORY
    LOCATION, AND (AFTER 'ENTER') THE LAST"
200 INPUTS,E
210 FOR A=S TO E
220 N=PEEK(A)
230 PRINTA;" ";
240 PRINTN;" ";
250 H=INT(N/16)
260 L=N-16*INT(N/16)
270 IF H>9 THEN GOSUB 330
280 IF L>9 THEN 350
290 IF H>9 THEN 310
300 PRINTCHR$(H+48);
310 PRINTCHR$(L+48);
320 GOTO 370
330 PRINTCHR$(H+55);
340 RETURN
350 PRINTCHR$(L+55);
360 GOTO 370
370 PRINT" ";CHR$(N)
380 W$=INKEY$
390 IFW$=" "THEN GOSUB420
400 NEXT
410 STOP
420 FOR DLAY=0 TO 200:NEXT
430 RETURN

```

# Petrometer

Ever wondered how much petrol your car is using? Well with PETROMETER you can calculate it easily. Just enter an initial milometer reading and then in turn new readings and the number of gallons used since the last reading. As a task you might like to see if you can change the program to produce miles per litre now that most petrol stations work in litres (see METRIC CONVERTER).

```
10 REM *****
20 REM * PETROMETER *
30 REM *****
40 CLS
50 PRINT"THIS PROGRAM CALCULATES"
60 PRINT"HOW MUCH PETROL YOU HAVE USED"
70 PRINT"YOU ENTER YOUR INITIAL MILAGE"
80 PRINT"AND THEN IN TURN THE LATEST"
90 PRINT"MILAGE RECORDING FOLLOWED BY THE"
100 PRINT"AMOUNT OF FUEL USED SINCE THE"
110 PRINT"LAST READING"
120 PRINT:PRINT:INPUT"INITIAL MILAGE";IM(T)
130 PRINT:INPUT"MILAGE NOW(BREAK TO END)"
    ;MN(T)
140 PRINT:INPUT"GALLONS USED";GU(T)
150 GOSUB 180
160 T=T+1
170 GOTO 130
180 CLS
190 PRINT"*****"
200 PRINT:PRINT"MILAGE";TAB(20);"M.P.G"
210 PRINT
220 FOR X=0 TO T
230 PRINTMN(X);"          ";
    :PRINT (MN(X)-IM(X))/GU(X)
240 IM(X+1)=MN(X)
250 NEXT X
260 RETURN
```

# Monitor

MONITOR allows you to enter machine code directly into the memory of your Dragon..You choose whether you wish to enter the byte in decimal or HEX and then just state where the first byte should be POKEd. To stop entry just enter 'S' instead of the next byte. It would be very easy to combine this program with HEXAPEEK and form a single quite powerful monitor for the Dragon. You could quite easily add a 'Byte modifier' routine to allow you to correct mistakes or make alterations to your machine code program.

```
10 REM *****
20 REM * MONITOR *
30 REM *****
40 CLS
50 PRINT"*****"
60 PRINT"          MONITOR"
70 PRINT"*****"
80 PRINT:PRINT"THIS PROGRAM WILL ALLOW
  YOU TO ENTER MACHINE CODE"
90 PRINT"DIRECTLY INTO MEMORY USING"
100 PRINT"EITHER DECIMAL OR HEX ENTRY"
110 INPUT"START ADDRESS FOR M/C";S
120 INPUT "DECIMAL (1) OR HEX(2) ENTRY";X
130 ON X GOTO 140,210
140 CLS
150 PRINT"ENTER BYTES SEPARATED BY 'ENTER'"
160 INPUT N$
170 IF LEFT$(N$,1)="S" THEN STOP
180 POKE S,VAL(N$)
190 PRINTS,PEEK(S)
200 S=S+1:GOTO 160
210 CLS
220 PRINT"ENTER BYTES IN HEX SEPARATED "
230 PRINT"BY 'ENTER'"
240 PRINT:INPUTN$
250 IF LEFT$(N$,1)="S" THEN STOP
260 A=ASC(LEFT$(N$,1))
```

```
270 B=ASC (RIGHT$(N$,1) )
280 IF A>63 THEN 370
290 H=16*(A-48)
300 IF B>63 THEN 390
310 L=B-48
320 N=H+L
330 POKE S,N
340 PRINTS,PEEK(S)
350 S=S+1
360 GOTO 240
370 H=16*(A-55)
380 GOTO 300
390 L=B-55
400 GOTO 320
```

# Metric Converter

With many measurements now going metric this program, METRIC CONVERTER may be a great help to you. It will either convert Gallons to Litres or Pounds to Kilos. The Gallons to Litre conversion part could be very useful in conjunction with PETROMETER now that many petrol stations work in metric. It would be quite easy for you to change a few lines so that the conversions were reversed, that is Litres to Gallons and Kilos to Pounds.

```
10 REM *****
20 REM * METRIC CONVERTER *
30 REM *****
40 CLS
50 PRINT"THIS PROGRAM WILL EITHER CONVERT
   GALLONS TO LITRES OR POUNDS TO KILOS"
60 PRINT"WHICH DO YOU WISH?(LITRES=1,
   KILOS=2)"
70 INPUT X
80 ON X GOTO 90,160
90 REM * GALLONS TO LITRES *
100 INPUT"VOLUME IN GALLONS";G
110 L=2.22*G
120 PRINTG;" GALLONS IS ";L;" LITRES"
130 INPUT "ANOTHER";Q$
140 IF LEFT$(Q$,1)="Y" THEN 100
150 STOP
160 REM * POUNDS TO KILOS *
170 INPUT "AMOUNT IN POUNDS";P
180 K=2.205*P
190 PRINTP;" POUNDS IS ";K;" KILOS"
200 INPUT "ANOTHER";Q$
210 IF LEFT$(Q$,1)="Y" THEN 170
220 STOP
```



# Wumpii

WUMPII is my version of an old favourite. It has been called many names such as WUMPUS or SUBMARINES. Essentially all you have to do is type in two numbers which represent a position on the game-board. The left hand number is entered first then (after pressing 'ENTER') the upper number is entered - so row first followed by column. If you are within one square of a WUMPUS then the Dragon will tell you so - and your object is to seek all six which Dragon has hidden somewhere in the grid. When you are successful he tells you so and prints a 'W' at the position. This can be a great game where players can take turns against the computer. It plays differently every time.

```
4 REM *****
5 REM * WUMPII *
6 REM *****
10 CLS
20 REM POSITION WUMPI
30 DIMWX(8)
40 DIMWY(8)
50 FOR J=1 TO 5
60 A=RND(8)
70 WX(J)=A
80 NEXT
90 FOR K=1 TO 5
100 A=RND(8)
110 WY(K)=A
120 NEXT
130 FOR A=1 TO (32*13)
140 PRINT@(A-1),CHR$(159);
150 NEXT A
160 PRINT@7,"1 2 3 4 5 6 7 8";
170 PRINT@(32*9)+7,"1 2 3 4 5 6 7 8";
180 FOR A=1 TO 8
190 PRINT@(32*A)+5,A;TAB(21);A;
```

```

200 NEXT
210 FOR DLAY=1 TO 200: NEXT:PRINT@(32*14)," "
215 PRINT@(32*13),"";:INPUT "MOVE(X,Y)";X,Y
220 GOSUB240
230 GOTO 210
240 REM WUMPUS HERE?
250 P=(X*32)+5+(2*Y)
260 FOR J=1 TO 5
270 IF (32*WX(J))+5+(2*WY(J))=P THEN 420
280 NEXT
290 GOTO 300
300 REM NO NONE HERE
310 REM BUT,NEAR BY?
320 P=(32*X)+5+(2*Y)
330 FOR J=1 TO 5
340 Q=(32*WX(J))+5+(2*WY(J))
350 IF Q=P+2 OR Q=P-2 THEN 450
360 IF Q=P+31 OR Q=P+32 OR Q=P+33 THEN 450
370 IF Q=P-31 OR Q=P-32 OR Q=P-33 THEN 450
380 NEXT
390 REM NO NONE NEAR
400 PRINT@(32*14),"NO WUMPII";
410 RETURN
420 PRINT@(32*14),"GOT HIM!"
430 PRINT@P,"W";
440 RETURN
450 PRINT@(32*14),"HE'S NEAR..."
460 RETURN

```

# Squash

Here is a version of SQUASH for the Dragon 32. You move your bat left and right using the '`<`' and '`>`' keys and bounce the ball off the walls. You have ten goes, and after ten misses your score is revealed. Good luck!

```
10 REM *****
20 REM * SQUASH *
30 REM *****
40 C=0
50 CLS
60 J=RND(10)+4:K=RND(15)+5
70 Q=-1
80 P=-1
90 FOR A=32 TO 63
100 PRINT@A,CHR$(147)
110 NEXT
120 FOR A=1 TO 14
130 PRINT@(32*A),CHR$(147);:PRINT@(32*A)
    +31,CHR$(147);
140 NEXT
150 X=10
160 REM
170 PRINT@(416)+X,CHR$(159)+CHR$(159);
180 A$=INKEY$
190 FOR DLAY=0 TO 50:NEXT
200 IF A$="," THEN GOSUB360
210 IF A$="." THEN GOSUB 370
220 PRINT@(32*J)+K," ";
230 K=K+(1*Q):J=J+(1*P)
240 PRINT@(32*J)+K,"O";
250 IF K>29 OR K<2 THEN Q=Q*-1
260 IF J<3 THEN P=P*-1
270 IF J=13 THEN 290
280 GOTO 160
290 IF (416+X)=(32*J)+K THEN 350
300 IF (418+X)=(32*J)+K THEN 350
310 IF (415+X)=(32*J)+K THEN 350
320 IF (417+X)=(32+J)+K THEN 350
```

```
330 C=C+1:IFC=10 THEN GOTO380
340 GOTO50
350 S=S+1:P=P*-1:GOTO160
360 PRINT@(416)+X," ";:X=X- 2:RETURN
370 PRINT@416+X," ";:X=X+2:RETURN
380 REM SCORE
390 CLS:PRINT"YOU SCORED ";S;" POINTS"
400 STOP
```

# Life

This is a version of John Conway's game LIFE .It depicts the evolution of a colony. The rules governing whether an individual will live, die or be born are very simple, but can give rise to surprisingly complicated outcomes. Essentially, every position on the 'board' is surrounded by eight other positions. If a cell has two or three neighbours then it lives. If there are three, and only three,neighbours then a new cell is born. Finally, any cell with more than three neighbours dies of overcrowding.

```
10 REM *****
20 REM * LIFE *
30 REM *****
40 CLS
50 DIMX(145)
60 DIMY(145)
70 Q=0
80 FOR A=1 TO 8
90 READ W
100 J(A)=W
110 NEXT
120 P=ASC("0")
130 W=32
140 FOR N=1 TO 12
150 FOR M=1 TO 12
160 X(N+10*M)=W
170 IF RND(100)>45 THEN X(N+10*M)=P
180 Y(N+10*M)=X(N+10*M)
190 NEXT:NEXT
200 Q=Q+1
210 FOR T=1 TO 12
220 FOR V=1 TO 12
230 S=T+10*V
240 IF Q=1 THEN 320
250 L=0
260 FOR Z=1 TO 8
270 IF X(S+J(Z)+1)=P THEN L=L+1
```

```
280 NEXT Z
290 IF X(S)=P AND L<>3 AND L<>2 THEN Y(S)=32
300 IF X(S)=32 AND L=3 THEN Y(S)=P
310 NEXT V: NEXT T
320 FOR E=1 TO 134
330 X(E)=Y(E)
340 NEXT E
350 PRINT@(32*3),"";
360 FOR G=1 TO 12
370 PRINTTAB(4);"";
380 FOR H=1 TO 12
390 S=G+10*H
400 PRINTCHR$(X(S));" ";
410 NEXT H
420 PRINT:NEXT G
430 PRINT@0+8,"GENERATION ";Q
440 GOTO 200
450 DATA 11,10,9,1,-1,-9,-10,-11
```

Also by the same author, and published by Century  
in association with 'Personal Computer World'

The Spectrum Handbook, £4.95

Also available from Century:

Microcomputing for Business:  
A User's Guide  
Edited by Dick Olney.

The Microcomputer Handbook  
A buyer's guide  
Edited by Dick Olney

Dictionary of New Information  
Technology  
by A.J.Meadows, M.Gordon  
and A.Singleton.

## CASSETTE

The programs in this book are available on a high quality cassette. Please send this box with a Cheque or P.O. for £2.95 to:

# SOFTTEK

329 Croxted Road, London  
SE24 9DB



Your search for programs for your  
Dragon 32 is over!

Here, for the first time, is an exciting  
collection of games and home applications.

These programs are simple to use  
and will make full use of the nine colour  
facilities of your Dragon.

This book will take you through Intergalactic  
Adventures, test your wits at Masterbrain  
and even help you budget your  
home accounts!

ISBN 0 7126 0173 2  
CENTURY PUBLISHING CO. LTD

£4.95